

NAG Library Routine Document

F02WGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F02WGF returns leading terms in the singular value decomposition (SVD) of a real general matrix and computes the corresponding left and right singular vectors.

2 Specification

```

SUBROUTINE F02WGF(M, N, K, NCV, AV, NCONV, SIGMA, U, LDU, V, LDV, RESID,
1 IUSER, RUSER, IFAIL)
    INTEGER          M, N, K, NCV, NCONV, LDU, LDV, IUSER(*), IFAIL
    double precision SIGMA(NCV), U(LDU,NCV), V(LDV,NCV), RESID(NCV),
1 RUSER(*)
    EXTERNAL         AV

```

3 Description

F02WGF computes a few, k , of the largest singular values and corresponding vectors of an m by n matrix A . The value of k should be small relative to m and n , for example $k \sim O(\min(m, n))$. The full singular value decomposition (SVD) of an m by n matrix A is given by

$$A = U \Sigma V^T,$$

where U and V are orthogonal and Σ is an m by n diagonal matrix with real diagonal elements, σ_i , such that

$$\sigma_1 \geq \sigma_2 \geq \cdots \geq \sigma_{\min(m,n)} \geq 0.$$

The σ_i are the *singular values* of A and the first $\min(m, n)$ columns of U and V are the *left* and *right singular vectors* of A .

If U_k , V_k denote the leading k columns of U and V respectively, and if Σ_k denotes the leading principal submatrix of Σ , then

$$A_k \equiv U_k \Sigma_k V_k^T$$

is the best rank- k approximation to A in both the 2-norm and the Frobenius norm.

The singular values and singular vectors satisfy

$$A v_i = \sigma_i u_i \quad \text{and} \quad A^T u_i = \sigma_i v_i \quad \text{so that} \quad A^T A v_i = \sigma_i^2 v_i \quad \text{and} \quad A A^T u_i = \sigma_i^2 u_i,$$

where u_i and v_i are the i th columns of U_k and V_k respectively.

Thus, for $m \geq n$, the largest singular values and corresponding right singular vectors are computed by finding eigenvalues and eigenvectors for the symmetric matrix $A^T A$. For $m < n$, the largest singular values and corresponding left singular vectors are computed by finding eigenvalues and eigenvectors for the symmetric matrix $A A^T$. These eigenvalues and eigenvectors are found using routines from Chapter F12. You should read the F12 Chapter Introduction for full details of the method used here.

The real matrix A is not explicitly supplied to F02WGF. Instead, you are required to supply a routine, AV, that must calculate one of the requested matrix-vector products Ax or $A^T x$ for a given real vector x (of length n or m respectively).

4 References

Wilkinson J H (1978) Singular Value Decomposition – Basic Aspects *Numerical Software – Needs and Availability* (ed D A H Jacobs) Academic Press

5 Parameters

- 1: M – INTEGER *Input*
On entry: m , the number of rows of the matrix A .
Constraint: $M \geq 0$.
 If $M = 0$, an immediate return is effected.
- 2: N – INTEGER *Input*
On entry: n , the number of columns of the matrix A .
Constraint: $N \geq 0$.
 If $N = 0$, an immediate return is effected.
- 3: K – INTEGER *Input*
On entry: k , the number of singular values to be computed.
Constraint: $0 < K < \min(M, N) - 1$.
- 4: NCV – INTEGER *Input*
On entry: the dimension of the arrays SIGMA and RESID and the second dimension of the arrays U and V as declared in the (sub)program from which F02WGF is called. This is the number of Lanczos basis vectors to use during the computation of the largest eigenvalues of $A^T A$ ($m \geq n$) or AA^T ($m < n$).
 At present there is no *a priori* analysis to guide the selection of NCV relative to K. However, it is recommended that $NCV \geq 2 \times K + 1$. If many problems of the same type are to be solved, you should experiment with varying NCV while keeping K fixed for a given test problem. This will usually decrease the required number of matrix-vector operations but it also increases the internal storage required to maintain the orthogonal basis vectors. The optimal ‘cross-over’ with respect to CPU time is problem dependent and must be determined empirically.
Constraint: $K < NCV \leq \min(M, N)$.
- 5: AV – SUBROUTINE, supplied by the user. *External Procedure*
 AV must return the vector result of the matrix-vector product Ax or $A^T x$, as indicated by the input value of IFLAG, for the given vector x .
 AV is called from F02WGF with the parameter IUSER and RUSER as supplied to F02WGF. You are free to use these arrays to supply information to AV.

The specification of AV is:

```
SUBROUTINE AV(IFLAG, M, N, X, AX, IUSER, RUSER)
  INTEGER      IFLAG, M, N, IUSER(*)
  double precision X(*), AX(*), RUSER(*)
```

- 1: IFLAG – INTEGER *Input/Output*
On entry: if $IFLAG = 1$, AX must return the m -vector result of the matrix-vector product Ax .
 If $IFLAG = 2$, AX must return the n -vector result of the matrix-vector product $A^T x$.

	<i>On exit:</i> may be used as a flag to indicate a failure in the computation of Ax or $A^T x$. If IFLAG is negative on exit from AV, F02WGF will exit immediately with IFAIL set to IFLAG.	
2:	M – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of rows of the matrix A .	
3:	N – INTEGER	<i>Input</i>
	<i>On entry:</i> the number of columns of the matrix A .	
4:	X(*) – double precision array	<i>Input</i>
	Note: the dimension of the array X is N if IFLAG = 1 and at least M if IFLAG = 2.	
	<i>On entry:</i> the vector to be pre-multiplied by the matrix A or A^T .	
5:	AX(*) – double precision array	<i>Output</i>
	Note: the dimension of the array AX is M if IFLAG = 1 and at least N if IFLAG = 2.	
	<i>On exit:</i> if IFLAG = 1, contains the m -vector result of the matrix-vector product Ax .	
	If IFLAG = 2, contains the n -vector result of the matrix-vector product $A^T x$.	
6:	IUSER(*) – INTEGER array	<i>User Workspace</i>
7:	RUSER(*) – double precision array	<i>User Workspace</i>

AV must be declared as EXTERNAL in the (sub)program from which F02WGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

- 6: NCONV – INTEGER *Output*
On exit: the number of converged singular values found.
- 7: SIGMA(NCV) – **double precision** array *Output*
On exit: the NCONV converged singular values are stored in the first NCONV elements of SIGMA.
- 8: U(LDU,NCV) – **double precision** array *Output*
On exit: the left singular vectors corresponding to the singular values stored in SIGMA.
The i th element of the j th left singular vector u_j is stored in $U(i, j)$, for $i = 1, 2, \dots, m$ and $j = 1, 2, \dots, NCONV$.
- 9: LDU – INTEGER *Input*
On entry: the first dimension of the array U as declared in the (sub)program from which F02WGF is called.
Constraint: $LDU \geq \max(1, M)$.
- 10: V(LDV,NCV) – **double precision** array *Output*
On exit: the right singular vectors corresponding to the singular values stored in SIGMA.
The i th element of the j th right singular vector v_j is stored in $V(i, j)$, for $i = 1, 2, \dots, n$ and $j = 1, 2, \dots, NCONV$.

- 11: LDV – INTEGER *Input*
On entry: the first dimension of the array V as declared in the (sub)program from which F02WGF is called.
Constraint: $LDV \geq \max(1, N)$.
- 12: RESID(NCV) – *double precision* array *Output*
On exit: the residual $\|Av_j - \sigma_j u_j\|$, for $m \geq n$, or $\|A^T u_j - \sigma_j v_j\|$, for $m < n$, for each of the converged singular values σ_j and corresponding left and right singular vectors u_j and v_j .
- 13: IUSER(*) – INTEGER array *User Workspace*
Note: the dimension of the array IUSER must be at least 1.
 IUSER is not used by F02WGF, but is passed directly to AV. It may be used to pass information to and from this routine.
- 14: RUSER(*) – *double precision* array *User Workspace*
Note: the dimension of the array RUSER must be at least 1.
 RUSER is not used by F02WGF, but is passed directly to AV. It may be used to pass information to and from this routine.
- 15: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
 F02WGF returns with IFAIL = 0 if at least k singular values have converged and the corresponding left and right singular vectors have been computed.

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL < 0

IFLAG was set to a negative value following a call to AV.

IFAIL = 1

On entry, $M < 0$.

IFAIL = 2

On entry, $N < 0$.

IFAIL = 3

On entry, $K < 0$.

IFAIL = 4

On entry, $NCV \leq K + 1$,
or $NCV > \min(M, N)$.

IFAIL = 5

On entry, $LDU < M$.

IFAIL = 6

On entry, $LDV < N$.

IFAIL = 7

No converged singular values were found to sufficient accuracy.

IFAIL = 8

The internal maximum number of iterations has been reached. Some singular values may have converged.

IFAIL = 9

IFAIL = 10

IFAIL = 20

An error occurred during an internal call. Consider increasing the size of NCV relative to K.

7 Accuracy

See Section 2.14.2 in the F08 Chapter Introduction.

8 Further Comments

None.

9 Example

This example finds the four largest singular values (σ) and corresponding right and left singular vectors for the matrix A , where A is the m by n real matrix derived from the simplest finite difference discretization of the two-dimensional kernel $k(s, t)dt$ where

$$k(s, t) = \begin{cases} s(t-1) & \text{if } 0 \leq s \leq t \leq 1 \\ t(s-1) & \text{if } 0 \leq t < s \leq 1 \end{cases}$$

9.1 Program Text

```
*
*   F02WGF Example Program Text
*   Mark 22 Release. NAG Copyright 2008.
*   .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5, NOUT=6)
      INTEGER          MAXM, MAXN, MAXNCV
      PARAMETER        (MAXM=500, MAXN=500, MAXNCV=25)
      INTEGER          LDU, LDV
      PARAMETER        (LDU=MAXM, LDV=MAXN)
*   .. Local Scalars ..
      INTEGER          I, IFAIL, K, M, N, NCONV, NCV
*   .. External Subroutines ..
      EXTERNAL         AV, F02WGF
*   .. Local Arrays ..
      DOUBLE PRECISION RESID(MAXNCV), RUSER(1), SIGMA(MAXNCV),
+                     U(LDU, MAXNCV), V(LDV, MAXNCV)
      INTEGER          IUSER(1)
```

```

*      .. Executable Statements ..
      WRITE (NOUT,*) 'F02WGF Example Program Results'
      WRITE (NOUT,*)
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N, K, NCV
      IF (N.LT.1 .OR. N.GT.MAXN) THEN
        WRITE (NOUT,99999) 'N is out of range: N = ', N
      ELSE IF (M.LT.1 .OR. M.GT.MAXM) THEN
        WRITE (NOUT,99999) 'M is out of range: M = ', M
      ELSE IF (NCV.GT.MAXNCV) THEN
        WRITE (NOUT,99999) 'NCV is out of range: NCV = ', NCV
      ELSE
*      Initialize for problem.
        IFAIL = 1
        CALL F02WGF(M,N,K,NCV,AV,NCONV,SIGMA,U,LDU,V,LDV,RESID,IUSER,
+          RUSER,IFAIL)
        IF (IFAIL.EQ.0) THEN
*
*      Print computed residuals
*
          WRITE (NOUT,*) ' Singular Value      Residual'
          WRITE (NOUT,99998) (SIGMA(I),RESID(I),I=1,NCONV)
        ELSE
          WRITE (NOUT,99997) IFAIL
        END IF
      END IF
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,F10.5,8X,G10.2)
99997 FORMAT (1X,' ** F02WGF Returned with IFAIL = ',I5)
      END
*
*      Matrix vector subroutines
*
      SUBROUTINE AV(IFLAG,M,N,X,AX,IUSER,RUSER)
*
*      Computes w <- A*x or w <- Trans(A)*x.
*
*      .. Parameters ..
      DOUBLE PRECISION ONE, ZERO
      PARAMETER      (ONE=1.0D+0,ZERO=0.0D+0)
*      .. Scalar Arguments ..
      INTEGER        IFLAG, M, N
*      .. Array Arguments ..
      DOUBLE PRECISION AX(*), RUSER(1), X(*)
      INTEGER        IUSER(1)
*      .. Local Scalars ..
      DOUBLE PRECISION H, K, S, T
      INTEGER        I, J
*      .. Intrinsic Functions ..
      INTRINSIC      DBLE, MIN
*      .. Executable Statements ..
      CONTINUE
      H = ONE/DBLE(M+1)
      K = ONE/DBLE(N+1)
      IF (IFLAG.EQ.1) THEN
        DO 20 I = 1, M
          AX(I) = ZERO
20      CONTINUE
        T = ZERO
*
        DO 80 J = 1, N
          T = T + K
          S = ZERO
          DO 40 I = 1, MIN(J,M)
            S = S + H
            AX(I) = AX(I) + K*S*(T-ONE)*X(J)
40      CONTINUE
          DO 60 I = J + 1, M
            S = S + H

```

```

        AX(I) = AX(I) + K*T*(S-ONE)*X(J)
60      CONTINUE
80      CONTINUE
      ELSE
        DO 100 I = 1, N
          AX(I) = ZERO
100     CONTINUE
        T = ZERO
*
        DO 160 J = 1, N
          T = T + K
          S = ZERO
          DO 120 I = 1, MIN(J,M)
            S = S + H
            AX(J) = AX(J) + K*S*(T-ONE)*X(I)
120     CONTINUE
          DO 140 I = J + 1, M
            S = S + H
            AX(J) = AX(J) + K*T*(S-ONE)*X(I)
140     CONTINUE
160     CONTINUE
        END IF
*
      RETURN
      END

```

9.2 Program Data

F02WGF Example Program Data
 100 500 4 10 : Values for M N K and NCV

9.3 Program Results

F02WGF Example Program Results

Singular Value	Residual
0.00830	0.19E-17
0.01223	0.23E-17
0.02381	0.18E-16
0.11274	0.12E-16
